```
/*******************************************************************************
* Project:       BRAC James P. Grant School of Public Health (JPGSPH)
* Study:         Master Pipeline - Round 1 & Round 2 Panel Survey
* Datasets:      Household Roster | Main Household Survey | Individual Survey
* Task:          Data Cleaning, Labeling, Transformation, Quality Assurance
*                and Panel Data Construction (R1 + R2)
* Author:        Ashiqur Rahman Rony
* Email:         ashiqurrahman.stat@gmail,com
* Last Updated:  17 April 2026
*
* PORTFOLIO / SAMPLE NOTE:
* This is a sample do-file created for demonstration and portfolio purposes.
* The full production version of this pipeline is significantly longer
* (several thousand lines) and includes extensive data quality checks (daily HFCs),
* algorithmic plausibility validations, ground-truth verifications from field
* monitors, and audio audits.
* The exhaustive list of line-by-line manual overrides, corrections, and detailed
* quality checks has been deleted/truncated here to keep the script concise
* and focused on overall architecture and logic.
*
* This sample was made more concise with the assistance of Gemini (xAI).
*******************************************************************************/

clear all
set more off


*===============================================================================
*===============================================================================
*** PART 1: ROUND 1 DATA CLEANING
*===============================================================================
*===============================================================================

*-------------------------------------------------------------------------------
* 1.1 MODULE 01: HOUSEHOLD ROSTER PROCESSING
*-------------------------------------------------------------------------------
* Importing csv data
import delimited "C:\Users\Rony\Desktop\Sample Do files\BHP hhroster survey (BRAC JPGSPH)_WIDE.csv", clear

*-------------------------------------------------------------------------------
* Variable Documentation: Labels, Value Sets, and Coding
*-------------------------------------------------------------------------------
** variable labeling
label variable a05 "Date time of Interview:"
label variable enum "Interviewer Name:"
label variable a06 "District Name:"
label variable a07 "Union Name:"
label variable a08 "Village Name:"
label variable a09 "Para Name"
label variable a10 "Landmark"
label variable a04 "Household ID No:"
label variable a03 "Interviewee Name:"
label variable consent "Consent"
label variable a02a "Upload digital signed consent paper"
label variable a13_1 "Respondent's Mobile Number-1 :"
label variable a13_2 "Respondent's Mobile Number-2 :"
label variable b1a "How many members have been sharing food in your household for the last 3 months?"
label variable hhroster_count "Repeat Count: Household Roster"
label variable line_b_1 "Line No. of Family Member"

forval i = 1/7 {
    * Basic Demographics
```

```
    lab var b01_`i' "Name of household member `i'"
    lab var b02_`i' "Relation with household head - member `i'"
    lab var b02oth_`i' "Other relation (specify) - member `i'"

    * Age, Gender and marital status
    lab var b04a_`i' "Age in completed years: member `i'"
    lab var b04b_`i' "Age in completed months: member `i'"
    lab var b05_`i' "Gender: member `i'"
    lab var b03_`i' "Marital status: member `i'"

    * Health and Disability
    lab var b06_`i' "Disability status (Multiple Response): member `i'"
    lab var b06oth_`i' "Other disability (specify): member `i'"

    * Education
    lab var b07a_`i' "Educational status: member `i'"
    lab var b07b_`i' "Highest class passed: member `i'"
    lab var b07c_`i' "Reasons for not attending school: member `i'"
    lab var b07coth_`i' "Other reasons for not attending (specify): member `i'"

    * Employment
    lab var b08_`i' "Profession (Multiple Response): member `i'"
}

** value sets and coding
label define a06 1 "Satkhira" 2 "Naogaon" // district
label define a07 1 "Kaikhali" 2 "Burigoalini" 3 "Paddopukur" 4 "Munshiganj" 5 "Chandan Nagar" ///
    6 "Hazi Nagar" 7 "Sreemantapur" 8 "Rasulpur" 97 "Others (Specify)" // Union Name

*------------------------------------------------------------------------------
* Cleaning & Open-Ended Translation
*------------------------------------------------------------------------------
/* The following 'Other (Specify)' responses were recorded in local language.
   Translations were done based on field supervisor notes and audio audits
   to ensure context remains intact for analysis. */

*--- Disability: b06oth_`i' ---
replace b06oth_1 = "Mental illness disability" if key=="uuid:db48473e-8c84-4c1d-bc50-f42ff247f22e"
replace b06oth_2 = "Born without toe" if key=="uuid:dc2c196b-0001-4b1c-ba2b-48d4529d4520"
replace b06oth_2 = "Speaks but unclear" if key=="uuid:6076b4bf-3edf-4665-a5c7-7cd93154d79e"
replace b06oth_3 = "Aged but not grown,Looks like 8 or 9-year-old" if key==
"uuid:40d1e01a-9a95-43e3-9a38-76349e7e534b"

*--- Education Dropout: b07coth_`i' ---
replace b07coth_1 = "Due to illness" if key=="uuid:4efd3c82-8262-4693-9eca-3cc194625daa"
replace b07coth_1 = "No suitable environment" if key=="uuid:563f3cc3-c2a9-458b-87d3-b48cf2c4d6cb"
replace b07coth_2 = "Father didn't want to educate her" if key==
"uuid:94c45a5c-f5f5-4c2a-a487-e2ab221ceafe"
replace b07coth_2 = "Stopped studying after father died" if key==
"uuid:f0f3dd9c-d68e-4409-ad4d-c28511137dd7"

/* Para names (sub-village units) were collected via open-ended text fields.
   Inconsistent naming and spelling variance consolidated here to ensure
   clean spatial grouping and merge-readability. */
replace a09 = trim(itrim(proper(a09))) // proper() handles Bengali-English transliteration casing;
itrim() removes internal double spaces.
replace a09 = "Abshason Para" if a09 == "Abshon Para"
replace a09 = "Adibashi Dighi Para" if inlist(a09, "Adibashi Dighi Para", "Adibashi Dighi Para")
replace a09 = "Adibashi Para" if inlist(a09, "Adibashipara", "Adibasi Para", "Adibashi Para")
replace a09 = "Ajogora Purnapara" if inlist(a09, "Ajogora Purbopara", "Ajugora Punnapara", "Ajugora")
replace a09 = "Ati Upor" if inlist(a09, "Atir Upor", "Atir Uppor")
replace a09 = "Boishkhali Koloni" if inlist(a09, "Boishkhali (Koloni Para)", "Boiskhali Koloni Para",
 "Boisokhali Koloni Para")
```

```

/* Consistency Check: Household Head (HHH) Validation
   Generates a binary flag: 1 if the household has exactly one head, 0 otherwise.
   In some cases enumerators mistakenly select more than one household head in a household or no household head in the household */
gen hhh_ok = 0
forval i = 1/8 {
    quietly replace hhh_ok = hhh_ok + 1 if b02_`i' == 1
}
replace hhh_ok = (hhh_ok == 1) // Convert count to binary: 1 = Valid (exactly one head), 0 = Error
label values hhh_ok d_yesno
label variable hhh_ok "Flag: Household has exactly one head"
tab hhh_ok, m // Quick summary for the log
drop hhh_ok

*--- Relation to Head (b02) ---
replace b02_3 = 11 if a04 == 22119 // Correcting relation per supervisor email
replace b02_4 = 11 if a04 == 22082 // Supervisor feedback: Member is a niece
replace b02_2 = 2 if a04 == 23025 // Correcting relation: Spouse, not child

* Duplicates id check
duplicates tag a04, gen(dup)
list a04 enum a05 a09 key if dup > 0, sepby(a04)
drop dup

* Logic Check: Child Marriage (Age < 18 & Not Unmarried)
/* Flagging households where any member under 18 is reported as married,
   widowed, or divorced (b03 != 1). */
di as error _newline "Potential Child Marriage Cases (Age <18 & Married)"
di as text "{hline 70}"
di "HHID (a04)" _col(15) "Member No" _col(30) "Name" _col(50) "Age" _col(60) "Status"
di "{hline 70}"
forval i = 1/8 {
    quietly count if b04a_`i' < 18 & b03_`i' != 1 & !missing(b03_`i')
    if r(N) > 0 {
        list a04 (b01_`i' b04a_`i' b03_`i') if b04a_`i' < 18 & b03_`i' != 1 & !missing(b03_`i'), clean noobs
    }
}
di "{hline 70}"

*--- Marital Status (b03) ---
replace b03_1 = 2 if a04 == 23021 // Verification confirmed member is married

*--- Variable Management (Dropping SurveyCTO Metadata) ---
/* Removing calculated variables, backend timestamps, and empty repeat group slots
   (slots 9-20) generated exclusively for SurveyCTO routing and auditing. */
drop audio_audit_mod_* start_* end_* du_t formdef_version starttime endtime ///
     duration district_lab union_lab village_lab a08_re v885-v892 v906-v913 ///
     b01_9-b01_20 b04a_9-b04a_20 b08_lab_* b08_l_* totalname_count

* Standardizing variable order for logical viewing
order a05, before(enum)

*--- Sample Attrition (Quality Control & Linkage) ---

* A. Informed Consent
drop if consent == 0

/* Due to poor data quality during initial visits, specific households were
   resurveyed. The initial corrupted submissions are dropped here via UUID. */
drop if inlist(key, "uuid:c3b2d24c-1ef8-4f27-b793-c2125564c491", ///
                    "uuid:9440d949-5b14-4bc9-acc2-d9c6d84f720d")
```

```

/* These households completed the roster but withdrew consent during the
   subsequent Main/Individual modules. Dropped for incomplete linkage. */
drop if inlist(a04, 11042, 11055, 22042, 23026)

/* Cases flagged and dropped due to poor data quality discovered
   during rigorous backend audio audits. */
drop if inlist(a04, 21028, 23002)

/* Study protocol requires 100% member coverage. These households
   are dropped because at least one member's individual data is missing. */
drop if inlist(a04, 11008, 14103)
save "Cfile01_Roster_Clean_Dataset.dta", replace


*--------------------------------------------------------------------------------
* 1.2 MODULE 02: HOUSEHOLD MAIN SURVEY PROCESSING
*--------------------------------------------------------------------------------
* Importing raw dataset
do "import_bhp_main.do"
use "Rfile02_Main_HH_Survey_Raw_Dataset.dta", clear

*--------------------------------------------------------------------------------
* Automated Variable & Value Labeling
*--------------------------------------------------------------------------------
/* Context: SurveyCTO's default import script is helpful, but it typically drops
   value labels for multiple-select and repeat group variables.

   Solution: Instead of writing hundreds of lines to manually label everything,
   I use the 'odksplit' package by ARCED Foundation. It reads the original XLSForm
   and dynamically applies all missing English labels to the dataset. */

cap noi ssc install odksplit // Installing dependencies
odksplit, survey("BHP_Main_surveyCTO+(Main+respondent-Household+level).xlsx") ///
          data("Rfile02_Main_HH_Survey_Raw_Dataset.dta") ///
          label(English) multiple single varlabel clear

*--------------------------------------------------------------------------------
* Advanced Data Transformation (Unpacking Repeat Groups)
*--------------------------------------------------------------------------------
/* The Problem: SurveyCTO repeat groups are great for field collection but messy
   for analysis. For example, in our 7-day food module (e101), questions about
   sourcing and pricing only appear for the foods the respondent actually selected.

   The Fix: To avoid writing complex, manual 'reshape' loops, I built a custom
   Stata package called 'exprep'. It algorithmically expands these nested repeat
   groups into a clean, analysis-ready wide format, generating new variable sets
   for each option while preserving the original labels.

   GitHub Repo: https://github.com/ashikpydev/exprep */

cap noi net install exprep, from("https://raw.githubusercontent.com/ashikpydev/exprep/main/")
* Automatically expanding the e101 food consumption repeat group
exprep, id(e101_repeat_id_*) base(e101) repvars(e101_source_* e101_total_price_*)

*--------------------------------------------------------------------------------
* Automated Back-Coding of "Other (specify)" Responses
*--------------------------------------------------------------------------------
/* The Problem: Every survey has open-ended "Other" fields. Standardizing,
   translating, and back-coding these responses usually requires hours of
   writing manual 'replace' commands, especially when checking for enumerator typos.

   The Fix: I developed a suite of Stata packages that automate this entire workflow.
```

```
   The packages scan the text responses, calculate frequencies, and if an "Other"
   response makes up more than 10% of the answers, it automatically generates a
   standardized code and merges it back into the main variable.

   Deployed Tools:
   - recode_oth (For single-select) -> https://github.com/ashikpydev/recode_oth
   - recode_nrep (For multiple-select) -> https://github.com/ashikpydev/recode_nrep
   - recode_rep (For repeat-groups) -> https://github.com/ashikpydev/recode_rep */

* Execute automated back-coding algorithms
recode_oth c02 c02oth // Single-select back-coding
recode_nrep g208 g208_97 g208oth // Multiple-select back-coding
recode_rep g1b6_3 g1b6_97_3 g1b6oth_3 // Nested repeat-group back-coding


save "Cfile02_Main_HH_Survey_Clean_Dataset.dta", replace


*------------------------------------------------------------------------------
* 1.3 MODULE 03: INDIVIDUAL SURVEY PROCESSING
*------------------------------------------------------------------------------
use "Rfile03_Individual_Survey_Raw_Dataset.dta", clear

*------------------------------------------------------------------------------
* Sample Attrition & Linkage Validation
*------------------------------------------------------------------------------
/* Context: In relational database design, individual-level modules must successfully
   link to the foundational household roster.

   Methodology: Implementing strict deterministic drops for unlinked records and incomplete household coverage.
   This enforces protocol compliance and ensures a robust, balanced panel for downstream analysis. */
* Drop orphaned records lacking a matching Household Roster entry
drop if inlist(a04, "13081", "13107", "21028", "21100", "23002", "24106")
* Drop households violating the "100% member coverage" study protocol
drop if inlist(a04, "11008", "14103")

*------------------------------------------------------------------------------
* Metadata Pruning & Memory Optimization
*------------------------------------------------------------------------------
/* The Problem: SurveyCTO generates massive amounts of backend variables, including
   audio audit markers, timestamps, and empty preloaded arrays.
   Leaving these variables in the dataset and slows down analytical processing.

   The Fix: Using wildcard (`*`) variable drops to cleanly and algorithmically
   prune all non-analytical metadata in a few lines, drastically reducing the
   dataset's memory footprint. */

drop v736 start_* end_* starttime endtime comments submissiondate ///
    audio_audit_mod_* preload_name_* preload_gender_* ///
    membername_count mem_sl_* mem_name_*
save "Cfile03_Individual_Clean_Dataset.dta", replace

*==============================================================================
*** PART 2: ROUND 1 RELATIONAL MERGING
*==============================================================================

/*=============================================================================
  DATA DELIVERY PROTOCOL:
  Per the study protocol, this script made three distinct analytical files
  from the cleaned modules:
  1. Person-Level Master (Individual Survey + G1 Module + Long Roster)
  2. Household-Level Master (Main HH Survey + Wide Roster)
```

```
  3. Master DATASET (Combined All datasets including household roster, household level, individual-level)
==============================================================================*/

*------------------------------------------------------------------------------
* 2.1 Normalizing Main HH Module (Section G1: Salinity Exposure)
*------------------------------------------------------------------------------
/* CONTEXT: Section G1 captures individual-level health and salinity exposure
   data, but it was embedded within the Household-level survey in a 'wide' array format.

   METHODOLOGY: To integrate this with the true individual-level surveys, we extract
   the G1 arrays, normalize them into a 'long' format, and dynamically generate
   a primary key (`resp_id`) for future 1:1 relational linkage. */

use "Cfile02_Main_HH_Survey_Clean_Dataset", clear
* Using wildcards (*) to cleanly isolate all G1 repeat group variables
keep a04 resp_id resp_line_num resp_name g1_r_count line_g1_* g1_* g1a* g1b*
reshape long line_g1_ line_g1_name_ line_g1_age_ age_month_ line_g1_gender_ line_g1_mstatus_ ///
             g1_ g1a1_ g1a1a_ g1a1b_ g1a1c_ g1a2_hour_ g1a2_min_ g1a3_ g1a3_1_ g1a3_2_ ///
             g1a3_3_ g1a3_97_ g1a3oth_ g1a4_ g1a5_ g1a5oth_ g1b1_ g1b1a_ g1b1b_ g1b1c_ ///
             g1b2_hour_ g1b2_min_ g1b3_ g1b4_ g1b4_1_ g1b4_2_ g1b4_3_ g1b4_4_ g1b4_97_ ///
             g1b4oth_ g1b5_ g1b6_ g1b6_1_ g1b6_2_ g1b6_3_ g1b6_4_ g1b6_5_ g1b6_6_ ///
             g1b6_97_ g1b6oth_ g1b7_hour_ g1b7_min_, i(a04) j(mid)

* Standardize nomenclature and generate linkage ID
ren line_g1_ sec_g1_line
ren line_g1_name_ sec_g1_name
ren line_g1_age_ sec_g1_age
ren age_month_ sec_g1_age_m
ren line_g1_gender_ sec_g1_gender
ren line_g1_mstatus_ sec_g1_mstatus
drop if sec_g1_line == ""
gen resp_id = a04 + sec_g1_line
drop mid resp_line_num resp_name g1_r_count
save "Section_G1_long", replace

*------------------------------------------------------------------------------
* 2.2 Normalizing Household Roster Demographics
*------------------------------------------------------------------------------
/* CONTEXT: The foundational demographic roster was processed in a wide format
   to accommodate household-level summary statistics.

   METHODOLOGY: We transpose these demographic arrays into a long panel structure
   to serve as the relational bridge linking Household and Individual data. */

use "Cfile01_Roster_Clean_Dataset", clear
* Utilizing wildcards to condense hundreds of explicit variable names
keep a04 consent line_b_* b0*
reshape long line_b_ b01_ b02_ b04a_ b04b_ b05_ b03_ b06_ b06_0_ b06_1_ b06_2_ b06_3_ ///
             b06_4_ b06_5_ b06_6_ b06_7_ b06_8_ b06_9_ b06_10_ b06_11_ b06_97_ b06oth_ ///
             b07a_ b07b_ b07c_ b07c_1_ b07c_2_ b07c_3_ b07c_4_ b07c_5_ b07c_6_ b07c_7_ ///
             b07c_8_ b07c_9_ b07c_10_ b07c_11_ b07c_97_ b07coth_ b08_ b08_1_ b08_2_ ///
             b08_3_ b08_4_ b08_5_ b08_6_ b08_7_ b08_8_ b08_9_ b08_10_ b08_11_ b08_12_ ///
             b08_13_ b08_14_ b08_15_ b08_16_ b08_17_ b08_18_ b08_19_ b08_20_ b08_21_ ///
             b08_22_ b08_23_ b08_24_ b08_25_ b08_26_ b08_27_ b08_28_ b08_29_ b08_30_ ///
             b08_31_ b08_32_ b08_33_ b08_34_ b08_35_ b08_36_ b08_37_ b08_38_ b08_39_ ///
             b08_40_ b08_41_ b08_42_ b08_43_ b08_44_ b08_45_ b08_46_ b08_47_ b08_48_ ///
             b08_49_ b08_50_ b08_51_ b08_52_ b08_53_ b08_54_ b08_55_ b08_56_ b08_57_ ///
             b08_58_ b08_59_ b08_60_ b08_61_ b08_62_ b08_63_ b08_64_ b08_99_, i(a04) j(mid)
drop if line_b_ == ""
gen resp_id = a04 + line_b_
drop mid
save "Roster_individual_information", replace
```

```
*------------------------------------------------------------------------------
* 2.3 DELIVERABLE 1 - PERSON-LEVEL MASTER DATASET
*------------------------------------------------------------------------------
/* CONTEXT: Integrating the standalone Individual survey with the newly normalized
   G1 module and the long-format demographic Roster. This yields a comprehensive,
   1:1 person-level analytical file spanning demographics, health, and exposures. */

use "Cfile03_Individual_Clean_Dataset", clear
* Merge 1: Main Individual Survey + G1 Module
merge 1:1 resp_id using "Section_G1_long"
drop sec_g1_line sec_g1_name sec_g1_age sec_g1_age_m sec_g1_gender sec_g1_mstatus _merge
* Merge 2: Append Demographic Roster
merge 1:1 a04 resp_id using "Roster_individual_information"
drop line_b_ _merge
* Pruning preloads and architecting logical variable flow
drop preload_name_* newmember_yn mem_list resp_name line_g1_name line_g1_age line_g1_gender
line_g1_mstatus line_g1_age_m
order b0* g1*, after(resp_id)
order consent, after(a04)

*--- Data Dictionary Metadata Injection ---
lab var b01_ "Name"
lab var b02_ "Relation with Household head to ${b01}"
lab var b04a_ "Age in completed year"
lab var b04b_ "Age in completed month"
lab var b05_ "Gender of ${b01}"
lab var b03_ "Marital Status of ${b01}"
lab var b06_ "Do ${b01} have any disability? [Multiple Responses]"
lab var b07a_ "Educational Status"
lab var b07b_ "Last class passed "
lab var b07c_ "Reasons for not attending or dropping out [Multiple Responses]"
lab var b08_ "Profession [Multiple Responses]"
lab var g1_ "Do you work in saltwater?"
lab var g1a1_ "Exposure to saline water in last 6 months?"
lab var g1b1_ "Exposure to intense heat or sunlight in last 6 months?"
lab var g1b5_ "Do you use any protective equipment?"
save "CMfile01_Combined_Individual_level_Dataset", replace

* Memory cleanup
cap erase "Section_G1_long.dta"
cap erase "Roster_individual_information.dta"

*------------------------------------------------------------------------------
* 2.4 DELIVERABLE 2 - HOUSEHOLD-LEVEL MASTER DATASET
*------------------------------------------------------------------------------
/* CONTEXT: Merging the baseline wide Roster back into the Main Household survey
   to create a unified, household-level analytical file capturing aggregate
   economics, assets, and comprehensive family structure. */

* Stage Wide Roster for Linkage
use "Cfile01_Roster_Clean_Dataset", clear
keep a04 consent line_b_* b0*
save "HH_roster_Information_wide", replace

* Prep Main HH Survey & Execute Linkage
use "Cfile02_Main_HH_Survey_Clean_Dataset", clear
* Drop wide G1 variables (already handled & normalized in Deliverable 1)
drop g1_1-g1b7_min_8
order line_g1_1-line_g1_mstatus_1, after(resp_name)
merge 1:1 a04 using "HH_roster_Information_wide"
order line_b_1-b08_99_8, after(line_g1_mstatus_1)
order consent, after(preload_landmark)
```

```

* Clean backend routing variables
drop preload_phone1 preload_phone2 preload_name_* newmember_yn mem_list line_g1_1-line_g1_mstatus_1 _merge
save "CMfile02_Combined_Household_level_Dataset", replace
cap erase "HH_roster_Information_wide.dta"

*--------------------------------------------------------------------------------
* 2.5 DELIVERABLE 3 - COMBINED MASTER DATASET
*--------------------------------------------------------------------------------
/* MERGE STRATEGY:
   Performing a many-to-one (m:1) relational join, connecting the master
   Household panel with the master Individual panel based on the Household ID.
   This yields the ultimate master dataset for complex, multi-level
   hierarchical modeling. */

use "CMfile02_Combined_Household_level_Dataset", clear
* droped redundant household-level spatial/telemetry variables to prevent inflation
drop a01 a06 preload_union preload_village preload_para preload_landmark ///
     membercount allmembers_r1 resp_id resp_line_num resp_name gpslatitude ///
     gpslongitude gpsaltitude gpsaccuracy formdef_version key
save "Combined_Household_level_Information_wide", replace

* Execute primary m:1 relational linkage
use "CMfile01_Combined_Individual_level_Dataset", clear
merge m:1 a04 using "Combined_Household_level_Information_wide"
drop _merge
save "CMfile03_Combine_All_Dataset", replace
cap erase "Combined_Household_level_Information_wide.dta"

*================================================================================
*** PART 3: MODULE 04 - PANEL DATA CONSTRUCTION (R1 + R2 MERGING)
*================================================================================

/*===============================================================================
  PANEL MERGING SECTION
  Round 2 datasets (Roster, Main HH, and Individual) have already been processed
  following the same structure as Round 1.
  This section appends R1 and R2, creates consistent panel structure, reshapes
  the roster to long format, and produces the final merged panel datasets.
===============================================================================*/

*--------------------------------------------------------------------------------
* 3.1 Merging Main Household Datasets (R1 + R2)
*--------------------------------------------------------------------------------
use "R2_Cfile02_Main_HH_Survey_Clean_Dataset", clear
preserve
    keep a04 a06 preload_union preload_village preload_para preload_landmark ///
         preload_phone1 preload_phone2
    save "dis", replace
restore
append using "Cfile02_Main_HH_Survey_Clean_Dataset"
drop mem_list

* Variable ordering for better readability
order e101 e101_1-e101_17 e101_repeat_count e101_repeat_id_* e101_*, before(e1a)
order b1 b09 b09_repeat_count b09_repeat_id_1 b09a_1 b09b_1 b09_repeat_id_2 ///
      b09a_2 b09b_2 b09_repeat_id_3 b09a_3 b09b_3 b09_repeat_id_4 b09a_4 b09b_4, ///
      after(a15a_15)
order a15a_5, after(a15a_4)
order rice_quantity pulse_quantity oil_quantity wheat_flour_semolina_quantity ///
      puffed_rice_flat_rice_quantity big_fish_quantity small_fish_quantity ///
      meat_quantity egg_quantity fruits_quantity milk_quantity ///
```

```
        leafy_vegetables_quantity non_leafy_vegetables_quantity salt_quantity ///
        spices_quantity betel_leaf_cig_etc_quantity onion_garlic_ginger_quantity ///
        rice_unit_price pulse_unit_price oil_unit_price wheat_flour_semolina_unit_price ///
        puffed_rice_flat_rice_unit_price big_fish_unit_price small_fish_unit_price ///
        meat_unit_price egg_unit_price fruits_unit_price milk_unit_price ///
        leafy_vegetables_unit_price non_leafy_vegetables_unit_price salt_unit_price ///
        spices_tot_price betel_leaf_cig_etc_unit_price onion_garlic_ginger_unit_price ///
        rice_source pulse_source oil_source wheat_flour_semolina_source ///
        puffed_rice_flat_rice_source big_fish_source small_fish_source ///
        meat_source egg_source fruits_source milk_source leafy_vegetables_source ///
        non_leafy_vegetables_source salt_source spices_source ///
        betel_leaf_cig_etc_source onion_garlic_ginger_source ///
        rice_freq pulse_freq oil_freq wheat_flour_semolina_freq ///
        puffed_rice_flat_rice_freq big_fish_freq small_fish_freq meat_freq ///
        egg_freq fruits_freq milk_freq leafy_vegetables_freq ///
        non_leafy_vegetables_freq salt_freq spices_freq ///
        betel_leaf_cig_etc_freq onion_garlic_ginger_freq, after(e101_freq_17)
order e201a e202a e203a, after(e208)
order c12_repeat_id_4 c12_repeat_name_4 c12a_4, after(c12a_3)
order r1_e201 r1_e202 r1_e203 r1_e204, after(e208)
order f112_repeat_id_3 f112_repeat_name_3 f112a_3 f112b_3 f113_3, after(f113_2)
order f116 f116_1 f116_2 f116_3 f116_4 f116_5 f116_6 f116_7 f116_97 f116oth, before(r2_f116)
order f122 f122oth, before(r2_f122)
order g201_9, before(g201_8)
order g3ba07 g3bb07, after(g3ba06)
save "RMfile01_R1_R2_Merged_Main_HH_Dataset", replace

*-----------------------------------------------------------------------------
* 3.2 Merging Individual Datasets (R1 + R2)
*-----------------------------------------------------------------------------
use "R2_Cfile03_Individual_Clean_Dataset", clear
merge m:1 a04 using "dis"
order a06 preload_union preload_village preload_para preload_landmark, after(a05)
drop _m
cap erase "dis.dta"
append using "Cfile03_Individual_Clean_Dataset"
drop mem_list
order g401a g401aoth g401b g401both g401c g401coth g401d g401doth g401e ///
      g401eoth g401f g401foth, before(g4a01a)
order h1a101_22 h1a101_23, after(h1a101_21)
order r1_h1a102 r1_h1a103 r1_h1a103_1-r1_h1a105e, before(h1a101_repeat_count)
order h2d01 h2d01_0 h2d01_1 h2d01_2 h2d01_3 h2d01_4, after(h2d01a)
order h2d05b3_98, after(h2d05b3_97)
order h304d_7, after(h304d_6)
order h5c07 h5c08 h5c08a h5c08b h5c08b_1-h5c09_5, after(preg_r2)
order h5e07a h5e07a_0-h5e07aoth, after(h5e06)
order h5f01 h5f02 h5f02_0-h5f03oth, after(h5e07aoth)
order h7b01 h7b02 h7b03 h7b04 h7b05, after(h7a10)
save "RMfile02_R1_R2_Merged_Individual_Dataset", replace

*-----------------------------------------------------------------------------
* 3.3 Merging Household-level with Individual-level (Full Panel)
*-----------------------------------------------------------------------------
use "RMfile02_R1_R2_Merged_Individual_Dataset", clear
merge m:m a04 round using "RMfile01_R1_R2_Merged_Main_HH_Dataset", force
drop _m preload_name_* mem_availibility_*
order resp_line_num religion religionoth a12 a14 a14_* a15 a15a a15a_* ///
      b1 b09 b09_* c01-c12a_4 buffaloes milk_cows_bulls goat_sheep ///
      chickens_ducks other_farm_animals pigeon d1a01-d1b05 brkdwn_expense ///
      d201-d203both e101-e208 r1_e201-r1_e204 e201a e202a e203a ///
      f101-f128 g1_r_count line_g1_* g1_* g201-g3e04, after(newmember_yn)
order preload_phone1 preload_phone2, after(preload_landmark)
order tank_cover reservoir_cover utensil_cover bottle_cover other_cover ///
```

```
      tank_duration reservoir_duration utensil_duration bottle_duration ///
      other_duration tank_photo reservoir_photo utensil_photo bottle_photo ///
      other_photo, after(f113_3)
save "RMfile03_R1_R2_HH_Merged_with_Individual_Dataset", replace

*-------------------------------------------------------------------------------
* 3.4 Merging & Reshaping Roster Information (R1 + R2) to Long Format
*-------------------------------------------------------------------------------
use "R2_Cfile01_Roster_Clean_Dataset", clear
append using "Cfile01_Roster_Clean_Dataset"
drop submissiondate a11latitude a11longitude a11altitude a11accuracy
order b1a, after(count_avail)
* Ordering R1 disability variables before R2
order r1_b06_1 r1_b06_0_1 r1_b06_1_1 r1_b06_2_1 r1_b06_3_1 r1_b06_4_1 ///
      r1_b06_5_1 r1_b06_6_1 r1_b06_7_1 r1_b06_8_1 r1_b06_9_1 r1_b06_10_1 ///
      r1_b06_11_1 r1_b06_97_1 r1_b06oth_1, before(b06_1)
forval i = 2/8 {
    order r1_b06_`i' r1_b06_0_`i' r1_b06_1_`i' r1_b06_2_`i' r1_b06_3_`i' ///
          r1_b06_4_`i' r1_b06_5_`i' r1_b06_6_`i' r1_b06_7_`i' r1_b06_8_`i' ///
          r1_b06_9_`i' r1_b06_10_`i' r1_b06_11_`i' r1_b06_97_`i', before(b06_`i')
}
keep round a06 a07 a08 a09 a10 a04 hhid identify_crrct_hh notidntify_crrct_hh_resn ///
     notidntify_crrct_hh_resnoth consent a02a newmem_in_hh newmem_in_hh_c b1a ///
     hhroster_count line_b_* b01_* new_qb0_* living_in_household_* b09a_* b09c_* ///
     q1a_* q1aoth_* b02_* b04a_* b04b_* b05_* b03_* r1_b06_* b06_* b06oth_* ///
     b07a_* b07b_* b07c_* b07coth_* b08_* b08oth_* key
reshape long line_b_ b01_ new_qb0_ living_in_household_ living_in_householdoth_ ///
             b09a_ b09c_ q1a_ q1aoth_ b02_ b04a_ b04b_ b05_ b03_ r1_b06_ ///
             r1_b06_0_ r1_b06_1_ r1_b06_2_ r1_b06_3_ r1_b06_4_ r1_b06_5_ ///
             r1_b06_6_ r1_b06_7_ r1_b06_8_ r1_b06_9_ r1_b06_10_ r1_b06_11_ ///
             r1_b06_97_ r1_b06oth_ b06_ b06oth_ b07a_ b07b_ b07c_ b07c_1_ ///
             b07c_2_ b07c_3_ b07c_4_ b07c_5_ b07c_6_ b07c_7_ b07c_8_ b07c_9_ ///
             b07c_10_ b07c_11_ b07c_97_ b07coth_ b08_ b08_1_ b08_2_ b08_3_ ///
             b08_4_ b08_5_ b08_6_ b08_7_ b08_8_ b08_9_ b08_10_ b08_11_ b08_12_ ///
             b08_13_ b08_14_ b08_15_ b08_16_ b08_17_ b08_18_ b08_19_ b08_20_ ///
             b08_21_ b08_22_ b08_23_ b08_24_ b08_25_ b08_26_ b08_27_ b08_28_ ///
             b08_29_ b08_30_ b08_31_ b08_32_ b08_33_ b08_34_ b08_35_ b08_36_ ///
             b08_37_ b08_38_ b08_39_ b08_40_ b08_41_ b08_42_ b08_43_ b08_44_ ///
             b08_45_ b08_46_ b08_47_ b08_48_ b08_49_ b08_50_ b08_51_ b08_52_ ///
             b08_53_ b08_54_ b08_55_ b08_56_ b08_57_ b08_58_ b08_59_ b08_60_ ///
             b08_61_ b08_62_ b08_63_ b08_64_ b08_97_ b08_99_ b08oth_, i(key) j(mid)
drop key mid round
drop if line_b_ == ""
drop if living_in_household_ != 1
replace line_b_ = "6" if line_b_ == "5" & a04 == "22114" & b01_ == "Jannatun begum"
gen resp_id = a04 + line_b_
drop hhid a06 a07 a08 a09 a10 newmem_in_hh newmem_in_hh_c consent a04
save "R1_R2_Roster_individual_information", replace

*-------------------------------------------------------------------------------
* 4.5 Final Merge: Roster Demographics into Full Panel Dataset
*-------------------------------------------------------------------------------
use "RMfile03_R1_R2_HH_Merged_with_Individual_Dataset", clear
merge m:m resp_id using "R1_R2_Roster_individual_information"
drop _merge

save "FINAL_PANEL_R1_R2_MASTER_DATASET", replace

*===============================================================================
* END OF MASTER PIPELINE SCRIPT
*===============================================================================
```